Solving and Plotting Ordinary Differential Equations in Maple

by Brent W. Woodruff - bw44197@appstate.edu

This file is available at http://www.fprimex.com under the 'Math' section.

Each section of this Maple file can be run completely independently of the other sections.

Run the 'Restart and Load the DEtools Package' section followed by the section on the topic you wish to view.

All variable names are 'unassigned' to minimize errors when more than one section is gone through at a time.

Conventions:

DE - The differential equation being solved or analyzed

DE1 - One of the DEs when working with a system of DEs

DE2 - The other DE when working with a system of DEs

IC - The initial Condition for a DE

IC1 - One of the ICs when more than one IC is required

IC2 - The other IC when more than one IC is required

Soln - Always represents the analytic solution to the DE or system of DEs

NumSoln - Numeric solution of the DE or system of DEs. A Maple procedure using a numerical method (default RK4).

EqSoln - Equilibrium solution to an autonomous differential equation

GenSoln - Specifies the general solution to the DE in question when looking at general solutions and particular solutions seperately

PartSoln - Specifies the particular solution to the DE in question when looking at general solutions and particular solutions seperately

k - spring constant when working with mass/spring problems

m - mass when working with mass/spring problems

f - The right hand side of an autonomous DE, considered as a function of its dependent variable. ($\frac{dy}{dt} = \mathrm{f}(y)$)

fprime - The derivative of f with respect to the dependent variable ($\frac{df}{dy}$)

BifY or BifS - The value of the dependent variable used for solving for the bifurcation value of the desired parameter

BifA, BifE1, or BifE2 - The value of the parameter at which the DE bifurcates

1 Restart and Load the DEtools Package

```
> restart;
> with(plots):
> with(DEtools):
```

```
Warning, the name changecoords has been redefined
```

2 Solving Differential Equations

2.1 General Solutions using dsolve

2.1.1 A Pure Time DE

Finding a general solution to $\frac{dy}{dt} = t^2 - t$

```
> DE := diff( y(t), t) = t^2 - t;
```

$$DE := \frac{\partial}{\partial t}\,\mathrm{y}(t) = t^2 - t$$

The solution to the DE is obtained using the dsolve command. Any constants that cannot be solved for are returned with an underscore preceding them. In this case _C1 is that constant.

```
> Soln := dsolve(DE, y(t) );
```

$$Soln := \mathrm{y}(t) = \frac{1}{3}\,t^3 - \frac{1}{2}\,t^2 + \_C1$$

```
> unassign('DE', 'Soln');
```

2.1.2 An Autonomous DE

Finding a general solution to $\frac{dy}{dt} = -(y-2)\,(y+1)$

```
> DE := diff(y(t), t) = -(y(t) - 2) * (y(t) + 1);
```

$$DE := \frac{\partial}{\partial t}\,\mathrm{y}(t) = -(\mathrm{y}(t) - 2)\,(\mathrm{y}(t) + 1)$$

```
> Soln := dsolve(DE, y(t));
```

$$Soln := \mathrm{y}(t) = \frac{2\,e^{(3\,t)}\,\_C1 + 1}{-1 + e^{(3\,t)}\,\_C1}$$

It is also easy to solve autonomous DEs for their equilibrium solutions.

```
> EqSoln := solve( rhs(DE)=0, 'y(t)' );
```

$$EqSoln := -1,\, 2$$

```
> unassign('DE', 'Soln', 'EqSoln');
```

2.1.3 A more complicated DE - u substitution

Finding a general solution to $\frac{dy}{dt} = -\frac{y}{t} + \frac{(t-1)\,y}{2}$

```
> DE := diff(y(t), t) = -y(t)/t + (t-1) / (2*y(t));
```

$$DE := \frac{\partial}{\partial t}\,\mathrm{y}(t) = -\frac{\mathrm{y}(t)}{t} + \frac{\frac{1}{2}\,(t-1)}{\mathrm{y}(t)}$$

```
> Soln := dsolve(DE, y(t) );
```

$$Soln := \mathrm{y}(t) = \frac{1}{6}\,\frac{\sqrt{9\,t^4 - 12\,t^3 + 36\,\_C1}}{t},\ \mathrm{y}(t) = -\frac{1}{6}\,\frac{\sqrt{9\,t^4 - 12\,t^3 + 36\,\_C1}}{t}$$

When there are multiple solutions to a differential equation and dsolve returns the answers seperated by commas, use the following command to access each one.

```
> Soln[1];
```

$$y(t) = \frac{1}{6}\frac{\sqrt{9\,t^4 - 12\,t^3 + 36\_C1}}{t}$$

```
> Soln[2];
```

$$y(t) = -\frac{1}{6}\frac{\sqrt{9\,t^4 - 12\,t^3 + 36\_C1}}{t}$$

To access the right hand side of a solution for plotting, evaluations, and the like use the 'right hand side' command.

```
> rhs(Soln[1]);
```

$$\frac{1}{6}\frac{\sqrt{9\,t^4 - 12\,t^3 + 36\_C1}}{t}$$

```
> rhs(Soln[2]);
```

$$-\frac{1}{6}\frac{\sqrt{9\,t^4 - 12\,t^3 + 36\_C1}}{t}$$

```
> unassign('DE', 'Soln');
```

2.1.4 A Second Order DE

Finding a general solution to $\frac{d^2 y}{dt^2} = -y$

```
> DE := diff( y(t), t$2) = -y(t);
```

$$DE := \frac{\partial^2}{\partial t^2}\,y(t) = -y(t)$$

```
> Soln := dsolve( DE, y(t) );
```

$$Soln := y(t) = \_C1\,\sin(t) + \_C2\,\cos(t)$$

```
> unassign( 'DE', 'Soln' );
```

2.2 Particular Solutions (Initial Value Problems) using dsolve

2.2.1 The same pure time DE with an initial condition (IC)

Finding a particular solution to $\frac{dy}{dt} = t^2 - t$, with initial condition $y(0) = 12$

```
> DE := diff( y(t), t) = t^2 - t ;
> IC :=           y(0) = 12      ;
```

$$DE := \frac{\partial}{\partial t}\,y(t) = t^2 - t$$

$$IC := y(0) = 12$$

```
> Soln := dsolve( {DE, IC}, y(t) );
```

$$Soln := y(t) = \frac{1}{3}\,t^3 - \frac{1}{2}\,t^2 + 12$$

```
> unassign( 'DE', 'IC', 'Soln' );
```

2.2.2 The same Autonomous DE with an IC

Finding a particular solution to $\frac{dy}{dt} = -(y-2)(y+1)$, with initial condition $\mathrm{y}(2) = 5$

```
> DE:= diff(y(t), t) = -(y(t) - 2) * (y(t) + 1) ;
> IC:=          y(2) = 5                        ;
```

$$DE := \frac{\partial}{\partial t}\,\mathrm{y}(t) = -(\mathrm{y}(t)-2)\,(\mathrm{y}(t)+1)$$

$$IC := \mathrm{y}(2) = 5$$

```
> Soln := dsolve( {DE, IC}, y(t));
```

$$Soln := \mathrm{y}(t) = \frac{4\,\frac{e^{(3\,t)}}{e^{6}} + 1}{-1 + \frac{2\,e^{(3\,t)}}{e^{6}}}$$

```
> simplify( Soln );
```

$$\mathrm{y}(t) = \frac{4\,e^{(3\,t-6)} + 1}{-1 + 2\,e^{(3\,t-6)}}$$

It is also easy to solve autonomous DEs for their equilibrium solutions.

```
> EqSoln := solve( rhs(DE)=0, 'y(t)' );
```

$$EqSoln := -1,\,2$$

```
> unassign('DE', 'IC', 'Soln');
```

2.2.3 The same complicated DE with an IC

Finding a particular solution to $\frac{dy}{dt} = -\frac{y}{t} + \frac{(t-1)\,y}{2}$, with initial condition $\mathrm{y}(1) = 4$

```
> DE := diff(y(t), t) = -y(t)/t + (t-1) / (2*y(t)) ;
> IC :=          y(1) = 4                          ;
```

$$DE := \frac{\partial}{\partial t}\,\mathrm{y}(t) = -\frac{\mathrm{y}(t)}{t} + \frac{\frac{1}{2}\,(t-1)}{\mathrm{y}(t)}$$

$$IC := \mathrm{y}(1) = 4$$

```
> Soln := dsolve( {DE, IC}, y(t) );
```

$$Soln := \mathrm{y}(t) = \frac{1}{6}\,\frac{\sqrt{9\,t^4 - 12\,t^3 + 579}}{t}$$

```
> unassign('DE', 'IC', 'Soln');
```

2.3 Higher order initial conditions using dsolve and D instead of diff

2.3.1 The same Second Order DE with two ICs

Finding a general solution to $\frac{d^2\,y}{dt^2} = -y$, with the initial conditions $\mathrm{y}(0) = 0$ and $\frac{\mathrm{dy}(0)}{dt} = 4$. The repeated composition operator, @@2, is used here to tell Maple to take two derivatives with the D operator.

```
> DE  := (D@@2)(y)(t) = -y(t) ;
> IC1 :=         y(0) =  0    ;
> IC2 :=      D(y)(0) = 4     ;
```

$$DE := (D^{(2)})(y)(t) = -y(t)$$

$$IC1 := y(0) = 0$$

$$IC2 := D(y)(0) = 4$$

```
> Soln := dsolve( {DE, IC1, IC2}, y(t) );
```

$$Soln := y(t) = 4\sin(t)$$

```
> unassign( 'DE', 'IC1', 'IC2', 'Soln' );
```

2.3.2 Throwing a ball off of a building

A ball is thrown off of a 100 meter building, upwards at an initial velocity of 10 meters per second. Let y=0 be ground level and y=100 be the height of the building. Assume gravity to be -9.8 meters per second squared. Find the height of the ball as a function of time.

```
> DE  := (D@@2)(p)(t)    =  -9.8 ;  # acceleration of the ball
> IC1 := p(0)            = 100   ;  # initial position of the ball
> IC2 := D(p)(0)         =  10   ;  # initial velocity of the ball
```

$$DE := (D^{(2)})(p)(t) = -9.8$$

$$IC1 := p(0) = 100$$

$$IC2 := D(p)(0) = 10$$

```
> GenSoln := dsolve( DE, p(t) );
```

$$GenSoln := p(t) = -\frac{49}{10}\,t^2 + \_C1\,t + \_C2$$

```
> PartSoln := dsolve( {DE, IC1, IC2}, p(t) );
```

$$PartSoln := p(t) = -\frac{49}{10}\,t^2 + 10\,t + 100$$

```
> unassign( 'DE', 'IC1', 'IC2', 'IC2', 'GenSoln', 'PartSoln' );
```

3 Plotting Solutions

3.1 dsolve for a solution, then plot it like any other function

Finding the solution to $\frac{dy}{dt} = y$, with the inital condition y(0) = 3

```
> DE := diff( y(t), t) = y(t) ;
> IC :=           y(0) = 3    ;
```

$$DE := \frac{\partial}{\partial t}\,y(t) = y(t)$$

$$IC := y(0) = 3$$

```
> EqSoln := solve( rhs(DE)=0, y(t) );
```

$$EqSoln := 0$$

```
> Soln := dsolve( {DE, IC}, y(t) );
```

$$Soln := y(t) = 3\,e^t$$

```
> plot( rhs(Soln), t=0..5, y=0..500);
```




```
> plot( { rhs(Soln), EqSoln }, t=0..5, y=0..500, thickness=3);
```




```
> unassign('DE', 'IC', 'Soln', 'EqSoln');
```

3.2 Numerical dsolve and odeplot

Finding numerical solution to $\frac{dy}{dt} = y$, with the inital condition y(0) = 3. Numerical solutions are required for the use of odeplot.

```
> DE := diff( y(t), t) = y(t) ;
> IC :=           y(0) = 3    ;
```

$$DE := \frac{\partial}{\partial t}\,\mathrm{y}(t) = \mathrm{y}(t)$$

$$IC := \mathrm{y}(0) = 3$$

The 'numeric' option is passed to dsolve which returns a Maple procedure that uses a numerical method. The default method is RK4, with others such as Euler availible. If the range of the independent variable is not specified, -10 to 10 is used.

```
> NumSoln := dsolve( {DE, IC}, y(t), numeric );
> odeplot(NumSoln);
```

$$NumSoln := \mathbf{proc}(rkf45\_x)\ \ldots\ \mathbf{end\ proc}$$



Here the same command is used with the addition of a range for the independent variable.

```
> NumSoln := dsolve( {DE, IC}, y(t), numeric, range=0..5 );
> odeplot(NumSoln);
```

$$NumSoln := \mathbf{proc}(rkf45\_x)\ \ldots\ \mathbf{end\ proc}$$




```
> unassign( 'DE', 'IC', 'NumSoln');
```

4 Other Plotting Techniques (Qualitative Analysis)

4.1 Plots for drawing Phase Lines of Autonomous DEs

Consider the autonomous DE $\frac{dy}{dt} = -(y-2)\,(y+1)$. I have declared it here the way DEs are normally declared when the use of dsolve is employed for finding the solution. The goal here is to plot f(y), where $\frac{dy}{dt} = \mathrm{f}(y)$.

```
> DE := diff(y(t), t) = -(y(t) - 2) * (y(t) + 1);
```

$$DE := \frac{\partial}{\partial t}\,\mathrm{y}(t) = -(\mathrm{y}(t) - 2)\,(\mathrm{y}(t) + 1)$$

In order to plot the right hand side as a function of y, the function y(t) needs to be written as the variable y.

```
> f := subs( y(t)=y, rhs(DE) );
```

$$f := -(y-2)\,(y+1)$$

```
> plot(f, y=-10..10, -10..10);
```




```
> unassign('DE', 'f');
```

4.2 Slope Fields with dfieldplot

Plotting the slope field for $\frac{dy}{dt} = -\frac{y}{t} + \frac{t-1}{2\,y}$

```
> DE := diff(y(t), t) = -y(t)/t + (t-1) / (2*y(t));
```

$$DE := \frac{\partial}{\partial t}\,\mathrm{y}(t) = -\frac{\mathrm{y}(t)}{t} + \frac{\frac{1}{2}\,(t-1)}{\mathrm{y}(t)}$$

```
> dfieldplot( DE, y(t), t=-10..10, y=-10..10);
```



The phaseportrait command will produce the field plot along with the plot of one or more particular solutions. Here, the initial conditions y(1) = 1 and y(1) = −1 are used.

```
>  phaseportrait( DE, y(t), t=-1..10, [[y(1)=1], [y(1)=-1]]);
```




```
>  unassign('DE');
```

4.3 Phase Portraits and Numerical Solutions For the Lotka-Volterra Model

Analyzing the Lotka Volterra Model defined as follows:

$\frac{dR}{dt} = 2\,R - 1.2\,R\,F$

$\frac{dF}{dt} = -F + .9\,R\,F$

Along with the initial conditions:

R(0) = 1

and F(0) = .5

```
> DE1 := D(R)(t) = 2*R(t) - 1.2*R(t)*F(t);
> DE2 := D(F)(t) =  -F(t) + 0.9*R(t)*F(t);
> IC1 := R(0) = 1.0;
> IC2 := F(0) = 0.5;
```

$$DE1 := \mathrm{D}(R)(t) = 2\,\mathrm{R}(t) - 1.2\,\mathrm{R}(t)\,\mathrm{F}(t)$$

$$DE2 := \mathrm{D}(F)(t) = -\mathrm{F}(t) + .9\,\mathrm{R}(t)\,\mathrm{F}(t)$$

$$IC1 := \mathrm{R}(0) = 1.0$$

$$IC2 := \mathrm{F}(0) = .5$$

Unfortunately, Maple cannot find an analytical solution to this system of ODEs

```
> Soln := dsolve( {DE1, DE2, IC1, IC2}, {R(t), F(t)} );
```

$$Soln :=$$

The dfieldplot command can be used to obtain a phase portrait without specifying any initial conditions.

```
> dfieldplot( {DE1, DE2}, [R(t), F(t)], t=0..10, R=0..3.5,
> F=0..4);
```



Using the phaseportrait command to plot the phase portrait along with the initial conditions.

```
> phaseportrait( {DE1, DE2}, {R(t), F(t)}, t=0..10, {[IC1, IC2]}
> );
```



An example of dfieldplot options for making your phase portraits nice and pretty.

```
>  dfieldplot( {DE1, DE2},
>  [R(t), F(t)],
>  t=0..10, R=0..3.5, F=0..4,
>  arrows=LARGE,
>  title=`Lotka-Volterra model`,
>  color=[rhs(DE1),rhs(DE2),.9]);
```



Though an analytical solution for the system is not possible, numerical solutions can still be obtained using dsolve with the numeric option. A number of different plots are possible using this numerical solution.

```
>  NumSoln := dsolve( {DE1, DE2, IC1, IC2}, {R(t), F(t)}, numeric,
>  range=0..10);
```

$$NumSoln := \mathbf{proc}(rkf45\_x)\ \ldots\ \mathbf{end\ proc}$$

Specifying R and F as the two axes produces the particular solution in a phase portrait manner.

```
>  odeplot( NumSoln, [R(t), F(t)] );
```



This method indicates to odeplot that the solutions should be plotted with respect to the independent variable t.

```
> odeplot( NumSoln, [[t, R(t)], [t, F(t)]] );
```



Producing a 3D representation of the solution curve. This is basically the particular solution plotted phase portrait style as above with the time axis added.

```
> odeplot( NumSoln, [t, R(t), F(t)] );
```




```
>  unassign( 'DE1', 'DE2', 'IC1', 'IC2', 'Soln', 'NumSoln' );
```

4.4 Visualization of the Mass and Spring problem

4.4.1 One second order DE

The physics for the mass and spring problem can be modeled using the second order ODE:

$$\frac{d^2 y}{dt^2} + \frac{k\,y}{m} = 0$$

where y is the position of the mass relative to its equilibrium position as a function of time, k is the spring constant from Hooke's Law, and m is the mass attached to the end of the spring. From the example given out of "Differential Equations" (Blanchard, Devaney, Hall), consider k and m such that $\frac{k}{m} = 1$.

```
>  k   := 1;                                # spring constant
>  m   := 1;                                # mass
>  IC1 :=                      y(0) = 1;    # initial position
>  IC2 :=                   D(y)(0) = 0;    # initial velocity
>  DE  := (D@@2)(y)(t) + (k/m)*y(t) = 0;
```

$$k := 1$$

$$m := 1$$

$$IC1 := \mathrm{y}(0) = 1$$

$$IC2 := \mathrm{D}(y)(0) = 0$$

$$DE := (\mathrm{D}^{(2)})(y)(t) + \mathrm{y}(t) = 0$$

Use dsolve to obtain the particular analytic solution.

```
> Soln := dsolve( {DE, IC1, IC2}, y(t) );
```

$$Soln := \mathrm{y}(t) = \cos(t)$$

```
> plot( rhs(Soln), t=0..4*Pi );
```




```
> unassign( 'k', 'm', 'IC1', 'IC2', 'DE', 'Soln' );
```

4.4.2 System of first order DEs

The physics for the mass and spring problem can be modeled using the second order ODE:

$$\frac{d^2 y}{dt^2} + \frac{k\,y}{m} = 0$$

where y is the position of the mass relative to its equilibrium position as a function of time, k is the spring constant from Hooke's Law, and m is the mass attached to the end of the spring. From the example given out of "Differential Equations" (Blanchard, Devaney, Hall), consider k and m such that $\frac{k}{m} = 1$.

This second order DE can be decomposed into a system of first order DEs as follows:

$$\frac{dy}{dt} = v$$

$$\frac{dv}{dt} = -y$$

Here, v is the velocity of the mass and is also a function of time.

```
>  k   := 1;               # spring constant
>  m   := 1;               # mass
>  IC1 := y(0) = 1;        # initial position
>  IC2 := v(0) = 0;        # initial velocity
>  DE1 := D(y)(t) =  v(t); # dy/dt
>  DE2 := D(v)(t) = -y(t); # dv/dt
```

$$k := 1$$

$$m := 1$$

$$\mathit{IC1} := \mathrm{y}(0) = 1$$

$$\mathit{IC2} := \mathrm{v}(0) = 0$$

$$\mathit{DE1} := \mathrm{D}(y)(t) = \mathrm{v}(t)$$

$$\mathit{DE2} := \mathrm{D}(v)(t) = -\mathrm{y}(t)$$

With just the two DEs defined a phase portrait can be created.

```
>  dfieldplot( {DE1, DE2}, [y(t), v(t)], t=0..4*Pi, y=-1..1,
>  v=-1..1);
```



The phaseportrait command can be used to generate a phase portrait along with the plot of one or more particular solutions.

```
>  phaseportrait( {DE1, DE2}, {y(t), v(t)}, t=0..4*Pi, {[IC1,
>  IC2]} );
```



In order to use odeplot for further qualitative analysis, a numerical solution is required.

```
>  NumSoln := dsolve( {DE1, DE2, IC1, IC2},
>  {y(t), v(t)},
>  numeric, range=0..4*Pi );
```

$$NumSoln := \mathbf{proc}(rkf45\_x)\ \ldots\ \mathbf{end\ proc}$$

Plotting the two particular solution curves: The position of the mass y(t)

, and the velocity of the mass v(t).

```
>  odeplot( NumSoln, [ [t,y(t)], [t,v(t)] ] );
```



The same particular solution plotted in phase portrait style. The scaling option allows for the spacing of the horizontal and vertical tickmarks to be the same.

```
>  odeplot( NumSoln, [ y(t), v(t) ], scaling=constrained );
```



A 3D visualization of the particular solution of the mass' position and velocity phase portrait style along the time axis.

```
>  odeplot( NumSoln, [ v(t), y(t), t ] );
```



Find the analytic solution to the system.

```
> Soln := dsolve( {DE1, DE2, IC1, IC2}, {y(t), v(t)} );
```

$$Soln := \{\mathrm{y}(t) = \cos(t),\, \mathrm{v}(t) = -\sin(t)\}$$

Two plots of the particular solutions, one showing maple's default plot and another using scaling options to more accurately represent the height of the sine and cosine curves.

```
> plot( [ rhs(Soln[1]), rhs(Soln[2]) ] , t=0..4*Pi);
> plot( [ rhs(Soln[1]), rhs(Soln[2]) ] , t=0..4*Pi,
> scaling=constrained);
```






```
> unassign( 'k', 'm', 'IC1', 'IC2', 'DE1', 'DE2', 'NumSoln', 'Soln'
> );
```

5 Bifurcations of Autonomous DEs

5.1 Bifurcation Values of a One Parameter Family

Find the bifurcation values of the one parameter family in the DE

$\frac{dy}{dt} = y^2 + 3\,y + a$

This declares the DE in a manner that can be passed to dsolve to obtain an analytical solution.

```
> DE := diff(y(t), t) = y(t)^2 + 3*y(t) + a;
```

$$DE := \frac{\partial}{\partial t}\,\mathrm{y}(t) = \mathrm{y}(t)^2 + 3\,\mathrm{y}(t) + a$$

Extract the right hand side of the DE, f(y), substituting the variable y for y(t) for the later diff and solve commands.

```
> f := subs( y(t)=y, rhs(DE) );
```

$$f := y^2 + 3\,y + a$$

The derivative of f(y).

```
> fprime := diff(f, y);
```

$$fprime := 2\,y + 3$$

Solving $\frac{df}{dy} = 0$ for y provides the values of y for which the bifurcation values occur.

```
> BifY := solve(fprime = 0, y);
```

$$BifY := \frac{-3}{2}$$

Plugging the above y value into f(y) and solving for a produces the parameter value where the bifurcation occurs. Use the subs command to do a simple substitution of variables as above. Use the eval command when the substitution involves a numerical value.

```
> BifA := solve( eval( f, y = BifY), a);
```

$$BifA := \frac{9}{4}$$

Plot f(y) with the parameter value $a = \frac{9}{4}$.

```
> plot( subs(a=BifA,   f), y=-5..5, F=-1..2 );
```



This creates and stores three plots of the DE using three different values of a: one just before the bifurcation occurs, one right at the bifurcation, and one after.

```
> Bifplots:=[plot( subs(a=BifA-1, f), y=-2..-1 ),
> plot( subs(a=BifA,   f), y=-2..-1 ),
> plot( subs(a=BifA+1, f), y=-2..-1 )]:
```

Display the three plots created side-by-side. Unfortunately, the display command ignores many view settings so it is difficult to get a good picture of the bifurcation using this method. Several "tricks" exist to create good arrayed plots, however the easiest thing to do is simply use the plot command like you normally would three seperate times.

```
> display( array(Bifplots) );
```



Create an animation of the f(y) as the value of the parameter changes. Click on the plot and then click the 'play' button which appears in a new toolbar at the top of your Maple window.

```
> animate(f, y=-5..5, a=-2..5, frames=50);
```




```
> unassign( 'DE', 'f', 'fprime', 'BifY', 'BifA', 'Bifplots' );
```

5.2 A more complicated example with unspecified parameters

Find the bifurcation values of the one parameter family in the DE

$\frac{dS}{dt} = k\,S\,(1 - \frac{S}{N})\,(\frac{S}{M} - 1) - E$

This declares the DE in a manner that can be passed to dsolve to obtain an analytical solution.

```
> DE := diff(S(t), t) = k*S(t)*(1 - S(t)/N) * (S(t)/M - 1) - E;
```

$$DE := \frac{\partial}{\partial t}\,\mathrm{S}(t) = \mathrm{S}(t)\,(1 - \frac{\mathrm{S}(t)}{N})\,(\frac{\mathrm{S}(t)}{M} - 1) - E$$

Extract the right hand side of the DE, f(S), substituting the variable S for S(t) for the later diff and solve commands.

```
> f := subs( S(t)=S, rhs(DE) );
```

$$f := S\,(1 - \frac{S}{N})\,(\frac{S}{M} - 1) - E$$

The derivative of f(S).

```
> fprime := diff(f, S);
```

$$\mathit{fprime} := (1 - \frac{S}{N})\,(\frac{S}{M} - 1) - \frac{S\,(\frac{S}{M} - 1)}{N} + \frac{S\,(1 - \frac{S}{N})}{M}$$

Solving $\frac{df}{dS} = 0$ for S provides the values of S for which the bifurcation values occur.

```
> BifY:= solve(fprime = 0, S);
```

$$\mathit{BifY} := \frac{1}{3}\,N + \frac{1}{3}\,M + \frac{1}{3}\,\sqrt{N^2 - N\,M + M^2},\ \frac{1}{3}\,N + \frac{1}{3}\,M - \frac{1}{3}\,\sqrt{N^2 - N\,M + M^2}$$

Plugging the above S value into f(S) and solving for E produces the parameter value where the bifurcation occurs. Use the subs command to do a simple substitution of variables as above. Use the eval command when the substitution involves a numerical value.

```
> BifE1 := simplify( solve( subs( S = BifY[1], f), E) );
```

$$\mathit{BifE1} := \frac{1}{27}(2\,N^3 - 3\,N^2\,M + 2\,N^2\,\sqrt{N^2 - N\,M + M^2} - 3\,N\,M^2 - 2\,N\,M\,\sqrt{N^2 - N\,M + M^2} + 2\,M^3 + 2\,M^2\,\sqrt{N^2 - N\,M + M^2})/(N\,M)$$

This time we have two y values for which bifurcations occur, so we repeat the above process for the second y value.

```
> BifE2 := simplify( solve( subs( S = BifY[2], f), E) );
```

$$\mathit{BifE2} := \frac{1}{27}(2\,N^3 - 3\,N^2\,M - 2\,N^2\,\sqrt{N^2 - N\,M + M^2} - 3\,N\,M^2 + 2\,N\,M\,\sqrt{N^2 - N\,M + M^2} + 2\,M^3 - 2\,M^2\,\sqrt{N^2 - N\,M + M^2})/(N\,M)$$

```
> unassign( 'DE', 'f', 'fprime', 'BifY', 'BifE1', 'BifE2' );
```